

TRS-80 Mailing List System

The TRS-80 Mailing List System is a RAM-based LEVEL II BASIC program that builds and maintains a list of names, and retrieves them individually.

When used with the TRS-80 Expansion Interface and Line Printer, it will print mailing labels on standard 3-1/2" by 15/16" label stock. Printing can be performed "selectively", either for a single name or for a group of names based upon zip codes, street names, cities, states, or even for a special "select code".

The program features a live keyboard for quick response times — you don't have to press **ENTER** when choosing an option. Another advantage is dynamic space allocation. This means the program will use as little memory as possible when storing names.

At the end of this manual, you'll find a complete listing of the Mailing List Program. It's provided as a convenience — in case you want to study the program, modify it, or verify that your CLOAded program in RAM is correct.

Note: This program requires 16K of RAM. It will not run if you attempt to protect memory with the MEMORY SIZE option.

To Load the Mailing List System

Prepare the tape recorder, volume setting 4-6, and press PLAY. Type:

CLOAD"M" **ENTER**

or just

CLOAD **ENTER**

When READY reappears, type:

RUN **ENTER**

Cassette tapes are used to save names and addresses. There is no limit to the number of cassette tapes you can create with the system, so theoretically you can have thousands of names in your system.

There is a limit to the number of names on each tape; it is a function of memory size.

RAM	Average Length of Entries	Capacity (Names Per Tape)
16K	50	150
32K	50	450
48K	50	750

How to Use the Mailing List System

The program offers 6 options:

```
<A>DD NAMES TO THE MAILING LIST
<R>EAD MAILING LIST FROM TAPE
<S>EARCH/CHANGE THE MAILING LIST
<L>IST NAMES ON THE SCREEN
<P>RINT MAILING LABELS
<W>RITE MAILING LIST ON TAPE
```

Whenever this option list ("menu") is displayed, you can select the desired option by typing the first letter of the command you wish to execute. For example, type the letter P to print mailing labels. You do not need to press **ENTER**

Add Names to the Mailing List (A-Option)

This is the first step in setting up the system. Before you begin, think a minute about how you will be using the mailing list, because the system allows for "selective label printing". This means you can choose to print labels only for people who live in a certain city or have a zip code beginning with certain digits. You can even use this "select code" as an indicator for information unrelated to the address.

Example: a club might use the select code to indicate whether or not a person has paid his membership dues. The club chooses to write PAID 78 in the select code field, to note that a member's dues have been paid through 1978. Using this as a select code, the club could then print labels for paid-up numbers to send them activities notices, and later print another set of labels for members-in-bad-standing to send them payment reminders.

When you're ready, type the letter A:

The program will ask:

WOULD YOU LIKE TO ENTER A SELECT CODE(Y/N)?

If you want to use this feature, reply Y. If you do not, reply N and that field will not be used. **You will only be given this select code option once for each RUN of the program.**

You can choose to include a telephone number or company name, but not both. The program asks for a T or a C; reply with the heading most appropriate for your application. You will only be given this select code option once for each RUN of the program.

The program asks if you want instructions for entering data. Reply Y or N.

Note: Before you begin, you should understand that the TRS-80, as all computers do, compares words and values by literal character by character evaluation. It doesn't care how you enter a city, for example, but it will look for it just as entered when you ask for labels. If the city is "Fort Worth", the following are *not* equal in the TRS-80's evaluation method:

FORT WORTH
FORT WORTH
FT. WORTH
FT WORTH

If you have entries using all four, and ask for labels for everyone in Fort Worth, you would have to make more than one run to get them all. Decide in the beginning how entries will be made and standardize all of them.

Note: Length of field is not controlled by the program. The maximum recommended field length for good labels is 26 characters.

Now you begin typing the names and addresses. Press **ENTER** after each field is typed in.

You may type the name two different ways. Either:
firstname lastname

NAME-----?JOHN SMITH
or: *last name, first name*

NAME-----?SMITH, JOHN

When labels are printed, the program will rearrange the name to print first name first and last name last. However, the names will be sorted according to the first name you type in. Therefore, type in all your names the same way. The presence of the comma will cause the switch during printout (so only use commas for *last name, first name* entries).

When finished, just press **ENTER** instead of typing a name. (If you run out of memory before all the names are in, a message **** OUT OF MEMORY **** will appear on the screen and you will have to create more than one tape.)

The names will be sorted automatically into alphabetical order (according to the first name you entered). This will take from a few seconds to several minutes.

Write Mailing List on Tape (W-Option)

The W option will take all names in memory and write them to tape. You should use this option after adding names to the mailing list or changing the mailing list. While you may write over old tapes, it is best to erase them with a high-quality bulk tape eraser such as the Radio Shack #44-210. The program will instruct:

RECORD MAILING LIST ON TAPE

SETUP THE RECORDER AND PRESS ENTER WHEN READY.
TO EXIT, PRESS E

Load a blank tape into the recorder. If the tape has a leader, be sure to advance the tape past the leader. Press PLAY and RECORD on the cassette recorder and press ENTER. The names will be recorded and when complete, you will be returned to the menu.

Read Names from Tape (R-Option)

Select this option to load a mailing list from tape (created with the Write option). Once you have created such tapes, you will normally use this option at the beginning of each session with the program.

Reading names from tape automatically clears out the mailing list previously stored in memory – so before Reading in a new list be sure to Write-to-tape any names you need to save.

Search/Change the Mailing List (S-Option)

This option lets you locate a particular name for reference or alteration. Type the letter S and the program will ask:

SEARCH/CHANGE

**TO DISPLAY A NAME, ENTER ENOUGH LETTERS TO IDENTIFY IT:
TO EXIT THIS MODE PRESS ENTER ?**

To locate John Smith's address type SMITH, JOHN (assuming you have entered it this way) and press **ENTER**. The program will display John Smith's name and address in seconds.

You're not required to enter the entire name. If you just type SMITH, the first SMITH in the list will appear. If you type S, the first name beginning with the letter S will appear. Once you're this close, it's easy to step through the list (type N) to find the desired name.

If the name you enter is not found, the message:

**** NOT IN THE LIST **** will appear.

There are 7 options listed above the name and address information.

P TO DISPLAY PREVIOUS NAME

Type the letter P (do not press **ENTER**) and the name stored immediately before the current name will be displayed.

N TO DISPLAY THE NEXT NAME

Type the letter N and the name stored immediately after the current will be displayed.

S TO SEARCH FOR A NEW NAME

Type the letter S and the program will respond just as above.

SHIFT D TO DELETE NAME

Hold down the shift key and type the letter D. The name displayed on the screen will be deleted from memory and the next name in the list will appear on the screen. Once deleted, a name cannot be recalled.

C TO CHANGE A FIELD

Type the letter C and the program will ask:

TYPE IN THE NUMBER OF THE FIELD & THEN ENTER THE NEW VALUE

The fields are numbered from 1 to 6 (1-7 if you use a select code). Type the number associated with the field you want to change (**do not** press **ENTER**). A question mark will appear beside the selected field. Now type in the new information and press **ENTER** when finished. The information on the screen will include your changes.

The program provides for error recovery with the following option.

Q TO CANCEL PREVIOUS CHANGES

If you have just made changes and would like to restore the original information because of entry errors, type the letter Q. This will return all the old information to the screen and clear all changes from memory for that one name. The Q option can only be used immediately after the C option. Once any other option (P,N,S,Shift D,E) has been used, recovery is impossible; reload the name tape or change the incorrect name entry.

E TO EXIT

Type E to return to the original menu.

List Names on the Screen (L-Option)

The L option lists the names in memory on the screen; no addresses are printed. Fourteen names are printed and the program asks:

PRESS E TO EXIT, ENTER TO CONTINUE

Type the letter E to return to the menu, or press **ENTER** to display the next fourteen names.

Print Mailing Labels (P-Option)

This option controls all printing operations. The program asks:

DO YOU WANT TO PRINT LABELS FOR THE WHOLE LIST(Y/N)?

If you reply N (no), the program asks:

WHICH FIELD WOULD YOU LIKE TO USE FOR A SELECT CODE(1 - 6)

(or 1-7 if the special select code was requested earlier). You can selectively print on any criterion. The numbers (1-6) refer to a field position:

- 1 is name
- 2 is company or telephone number
- 3 is address
- 4 is city
- 5 is state
- 6 is zip code
- 7 (if used) is a special select code.

Example:

You choose field 4 (city) as a select code and the value you type is: FT. WORTH. The program will print mailing labels for all names which have FT. WORTH as the city. If you type only FT. W as the select code value, the program may print labels for FT. WORTH, FT. WAYNE, and FT. WALTON BEACH. This feature is especially useful when applied to the zip code field. By specifying the first 3 digits of the zip code, you can print labels for "sorted bundles" and take advantage of reduced postal rates.

Before printing begins, several other questions appear on the screen. If you use COMPANY NAME instead of TELEPHONE NUMBER, you have this option:

DO YOU WANT TO PRINT THE COMPANY NAME ON THE LABELS(Y/N)?

"Y" prints the name, "N" does not.

Since the TRS-80 line printer has adjustable character sizes, you are given a trial run.

DO YOU WANT A TRIAL RUN(Y/N)?

If you reply Y, the first label will be printed 3 times. While the printer is working you should adjust the character size and align the labels so the address is well centered. You are given as many trial runs as necessary.

Once the labels are aligned, reply "N" for "no more trial runs" and the program will instruct:

PRESS ENTER TO START PRINTING, E TO EXIT

Just press **ENTER** and the labels will be printed. You may also type E to return to the menu. When printing is completed, the program asks:

DO YOU WANT TO READ IN ANY MORE TAPES(Y/N)?

This feature allows you to load additional names into memory and continue printing with the same selection criterion. A reply of N (no) returns you to the menu.

Note: If you choose to load an additional tape, you must print the labels for the names on this additional tape. To load another tape without printing, or with a different select code, you must first return to the menu, replying "NO", and load the new tape with the R option.

Program Listing

```
0 CLS:PRINT@276, "*** INITIALIZING ***":CLEAR(0)
1 REM -(C)1978 BY RADIO SHACK, A DIV. OF TANDY CORP. -ALL RIGHTS RESERVED.
5 IFMEM>40000THENCLEAR(31500):HI=401
   ELSEIFMEM>20000THENCLEAR(18800):HI=251
   ELSECLEAR(6000):HI=81
10 DEFINITI-N, G, E, P, S:DIMFI$(HI, 6)
20 HI=HI-1:N=0:M=0
30 POKE16553, 255:RESTORE:FORI=1TO6:READB$(I):NEXT:CLS:PRINT@17, "MAILING LIST SYSTEM":
   PRINT@131, "OPTIONS AVAILABLE:"
40 PRINT@256, "<A>DD NAMES TO THE MAILING LIST":PRINT"<R>EAD MAILING LIST FROM TAPE"
50 PRINT"<S>EARCH/CHANGE THE MAILING LIST":PRINT"<L>IST NAMES ON THE SCREEN"
60 PRINT"<P>RINT MAILING LABELS":PRINT"<W>RITE MAILING LIST ON TAPE"
65 PRINT@715, "*** # OF NAMES IN LIST=";N;" ***"
70 PRINT@832, "SELECT YOUR OPTION"
80 P=851:PRINT@P, CHR$(143)
90 GOSUB3000:I=1
100 IFB$(I)=A$THEN110:ELSEI=I+1:IFI<7THEN100:ELSE90
110 ON I GOTO 120, 800, 900, 600, 2000, 700
120 IFM<>0THEN295
140 CLS:PRINT@17, "ADD NAMES TO THE MAILING LIST"
150 PRINT@128, "WOULD YOU LIKE TO ENTER A SELECT CODE(Y/N)?":P=172:GOSUB3000
160 IFA$="Y"THENPRINT"Y":M=6:ELSEPRINT"N":M=5
170 PRINT@256, "YOU HAVE THE OPTION TO INCLUDE TELEPHONE # OR COMPANY NAME IN"
180 PRINT"YOUR MAILING LIST. WHICH DO YOU SELECT(T/C)?":P=365:G=0
190 GOSUB3000:IFA$="T"THENG=1:GOTO200:ELSEIFA$<>"C"THEN190
200 PRINTA$:PRINT@448, "DO YOU WANT INSTRUCTIONS ON HOW TO ENTER DATA(Y/N)?":P=500
210 GOSUB3000:IFA$="N"THEN300
220 CLS:GOSUB5000
230 PRINT@512, "REPRODUCED ABOVE IS THE SCREEN AS IT WILL APPEAR DURING DATA"
240 PRINT"ENTRY. TYPE IN EACH ITEM OF INFORMATION FOR A GIVEN PERSON AND"
250 PRINT"PRESS ENTER AFTER YOU HAVE ENTERED THE LAST ITEM, YOU WILL BE"
260 PRINT"PROMPTED IN THE SAME WAY FOR INFORMATION FOR THE NEXT PERSON. "
270 PRINT"  YOU CAN EXIT THIS MODE BY PRESSING ENTER (WITHOUT TYPING"
280 PRINT"ANYTHING) WHEN THE COMPUTER IS WAITING FOR THE NAME. "
290 PRINT@989, "":GOSUB6000:P=1013:GOSUB3000
295 E=0
300 CLS:IFN<HITHEIN310:ELSEIFE=1THENGOSUB9000
305 GOTO400
310 N=N+1:I=0:GOSUB5000
315 PRINT"":PRINT"":PRINT"  ":FORK=0TO55:PRINT"-":NEXT:PRINT"":PRINT"":PRINT"  TO
   EXIT THIS MODE PRESS ENTER. ELSE TYPE IN THE DATA":K=15
320 PRINT@K, "?":P=16:GOSUB4000:FI$(N, I)=B$(I):IFI<>0THEN330:ELSEIFFI$(N, I)<>""THEN33
0
```

(Program Listing, cont.)

```
322 N=N-1:IFE=0THEN30:ELSEGOSUB9000:GOTO30
330 I=I+1:K=K+64:IFI<=MTHEN320:ELSEE=1:GOTO300
400 PRINT@148,"** OUT OF MEMORY **":PRINT@256,"DO YOU WANT TO RECORD THE CURRENT LIST
    ON TAPE(Y/N)?":P=309:GOSUB3000
410 PRINT"":PRINT"":IFA$="Y"THENGOSUB10000
420 PRINT"DO YOU WANT TO CLEAR THE MEMORY FOR ADDITIONAL NAMES(Y/N)?":GOSUB8000
430 IFA$="Y"THENN=0
440 GOTO30
600 CLS:IFN=0THENGOSUB11000:GOTO30
610 I=1
620 J=1
630 PRINTFI$(I,0):J=J+1:I=I+1:IFI>NTHEN650:ELSEIFJ<15THEN630
640 PRINT"":PRINT"    PRESS E TO EXIT, ENTER TO CONTINUE":GOSUB8000:IFASC(A$)=69THEN
    30ELSEPRINT"":GOTO620
650 PRINT"    ** END OF LIST **":GOSUB6000:GOSUB8000:GOTO30
700 CLS:IFN=0THEN GOSUB11000:GOTO30
710 PRINT@17,"RECORD MAILING LIST ON TAPE":PRINT"":GOSUB10000:GOTO30
800 CLS:PRINT@17,"READ MAILING LIST":GOSUB14000:GOTO30
900 E=0
905 IFE=1THENGOSUB16000:E=0
906 CLS:IFN=0THENGOSUB11000:GOTO30
910 PRINT@20,"SEARCH/CHANGE"
920 PRINT@128,"TO DISPLAY A NAME, ENTER ENOUGH LETTERS TO IDENTIFY IT."
930 PRINT"TO EXIT THIS MODE PRESS ENTER ?":
940 I=0:P=31:GOSUB4000:IFB$(0)=" "THEN30
950 J=1:K=N
960 IFK<JTHENPRINT@223,"":CHR$(31):PRINT@394,"** ";B$(0); " NOT IN THE LIST **":PRINT@
    223,"":GOTO940
970 I=INT((J+K)/2):B$(1)=LEFT$(FI$(I,0),LEN(B$(0)))
980 IFB$(0)<B$(1)THENK=I-1:GOTO960
990 IFB$(0)>B$(1)THENJ=I+1:GOTO960
1000 CLS:PRINT@4,"TYPE:":PRINT@128,"<P> TO DISPLAY PREVIOUS NAME"
1010 PRINT@160,"<SHIFT D> TO DELETE NAME":PRINT@192,"<N> TO DISPLAY NEXT NAME"
1020 PRINT@224,"<C> TO CHANGE A FIELD":PRINT@256,"<S> TO SEARCH FOR A NEW NAME"
1030 PRINT@288,"<Q> TO CANCEL PREVIOUS CHANGES":PRINT@343,"<E> TO EXIT"
1040 FORJ=0TOM:B$(J)=FI$(I,J):NEXT
1045 PRINT@512,"":CHR$(31):
1050 GOSUB17000:K=512:FORJ=0TOM:PRINT@K,J+1:PRINT@K+9,B$(J):K=K+64:NEXT
1060 P=387:GOSUB3000:IFA$<"C"THEN1130
1070 PRINT@387,"TYPE IN THE NUMBER OF THE FIELD & THEN ENTER THE NEW VALUE":
1080 P=446:GOSUB3000:IFA$<"1"ORA$>"7"THEN1080ELSEJ=VAL(A$)
1090 IFJ>M+1THEN1080ELSEPRINT@387,"":CHR$(30):J=J-1
1100 K=547+64*J:PRINT@K,"?":K=I:I=J:P=36:FI$(0,0)=B$(J):GOSUB4000:I=K
```

(Program Listing, cont.)

```
1110 IFB$(J)="" THEN B$(J)=FI$(0,0)
1120 GOTO1045
1130 IFA$="Q" THEN1040
1135 IFA$<>"E" THEN1140
1136 GOSUB15000: IFE=1 THEN GOSUB16000
1137 GOTO30
1140 IFA$="S" THEN GOSUB15000: GOTO905
1150 IFA$<>"N" THEN1160 ELSE GOSUB15000
1151 I=I+1: IF I>N THEN905 ELSE IFFI$(I,0)="" THEN1151 ELSE1040
1160 IFA$<>"P" THEN1170 ELSE GOSUB15000
1161 I=I-1: IF I<1 THEN905 ELSE IFFI$(I,0)="" THEN1161 ELSE1040
1170 IF ASC(A$)<>100 THEN1060 ELSE FFI$(I,0)="" : E=1
1180 I=I+1: IF I>N THEN905 ELSE IFFI$(I,0)="" THEN1180 ELSE1040
2000 CLS: IF N=0 THEN GOSUB11000: GOTO30
2010 PRINT@18, "PRINT MAILING LABELS"
2020 PRINT@128, "DO YOU WANT TO PRINT LABELS FOR THE WHOLE LIST(Y/N)?":
2030 GOSUB8000: IFA$="Y" THEN B$(0)="" : PRINT"Y": PRINT"": GOTO2070 ELSE PRINT"N":
2040 PRINT"": PRINT"": PRINT"WHICH FIELD WOULD YOU LIKE TO USE FOR A SELECT CODE(1 -"; M
+1; ")":
2050 P=316: GOSUB3000: IFA$<"1" OR A$>"7" THEN2050 ELSE S=VAL(A$)-1
2060 IF S>M THEN2050 ELSE PRINTS+1: PRINT"": PRINT"TYPE IN A VALUE FOR THE SELECT CODE?":
P=36: I=0: GOSUB4000: PRINT"": PRINT""
2070 IF G=1 THEN E=0: GOTO2090: ELSE PRINT"DO YOU WANT TO PRINT THE COMPANY NAME ON THE LAB
ELS(Y/N)?": GOSUB8000
2080 IFA$="N" THEN E=0 ELSE E=1
2090 CLS: PRINT"YOU WILL NOW BE GIVEN TRIAL RUNS TO ALIGN THE LABELS": PRINT"AND ADJUST
CHARACTER SIZE. "
2100 PRINT"A TRIAL RUN CONSISTS OF PRINTING 3 LABELS. "
2110 PRINT"AFTER EACH TRIAL RUN YOU MAY ASK FOR ANOTHER TRIAL RUN"
2120 PRINT"OR GO AHEAD AND PRINT THE LABELS. "
2130 PRINT"SETUP THE LABELS AND TURN ON THE PRINTER.
2135 PRINT"": PRINT"DO YOU WANT A TRIAL RUN(Y/N)?": GOSUB8000: PRINT"": PRINT"": IFA$="N"
THEN2160
2140 IF PEEK(14312)>127 THEN PRINT"** PRINTER NOT READY **": GOTO2135
2150 FOR J=1 TO 3: I=1: GOSUB19000: NEXT: GOTO2135
2160 PRINT"PRESS ENTER TO START PRINTING, E TO EXIT": GOSUB8000: PRINT"": IFA$="E" THEN 3
0
2170 IF PEEK(14312)>128 THEN PRINT"": PRINT"** PRINTER NOT READY **": PRINT"": GOTO2160
2175 I=1
2180 A$="": J=1
2190 A$=INKEY$: IFA$<>" " THEN2200 ELSE J=J+1: IF J<5 THEN2190 ELSE2210
2200 PRINT"": PRINT"PRESS ENTER TO CONTINUE, E TO EXIT": GOSUB8000: PRINT"": IFA$="E" THE
N30
```

(Program Listing, cont.)

```
2210 GOSUB18000:I=I+1:IFI<=NTHEN2180
2220 PRINT"":PRINT"DO YOU WANT TO READ IN ANY MORE TAPES(Y/N)?":GOSUB8000:PRINT""
2230 IFA$="Y"THENGOSUB14000:IFA$="E"THEN30ELSEPRINT"":PRINT"":GOSUB6000:GOSUB8000:PRI
    NT"":PRINT"":GOTO2170
2240 GOTO30
3000 A$=""
3010 PRINT@P,CHR$(143):L=1
3020 A$=INKEY$:IFA$<>""THEN PRINT@P," ":RETURN
3025 L=L+1:IFL<16THEN3020
3030 PRINT@P," ":FORL=1TO10:NEXT:GOTO3010
4000 PRINTCHR$(14):B$(I)=""
4010 A$=INKEY$:IFA$=""THEN4010
4020 IFA$=CHR$(13)THENPRINTCHR$(15):RETURN
4030 IFA$=CHR$(8)THENIFB$(I)<>""THENPRINTA$:B$(I)=LEFT$(B$(I),LEN(B$(I))-1):GOTO4010
    :ELSEGOTO4010
4035 IFA$=CHR$(24)THENPRINTCHR$(29):STRING$(P,CHR$(25)):CHR$(30):GOTO4000
4040 IFASC(A$)<32OR ASC(A$)>90THEN4010
4050 PRINTA$:IFA$=","THENA$=","
4060 B$(I)=B$(I)+A$:GOTO4010
5000 PRINT"NAME-----"
5010 IFG=1THENPRINT"TELEPHONE-----":ELSEPRINT"COMPANY-----"
5020 PRINT"ADDRESS-----":PRINT"CITY-----":PRINT"STATE-----"
5030 PRINT"ZIP-----":IFM=5THEN RETURN
5040 PRINT"SELECT CODE----":RETURN
6000 PRINT"PRESS ENTER TO CONTINUE":RETURN
7000 PRINT"** NO DATA IN MEMORY **":RETURN
8000 GOSUB13000:GOSUB3000:RETURN
9000 IFN<2THEN RETURN
9005 CLS:PRINT@276,"** SORTING LIST **"
9010 I=0:FORJ=2TON
9020 K=J-1:FORL=0TOM:B$(L)=FI$(J,L):NEXT
9030 IFB$(I)>=FI$(K,I)THEN9060
9040 FORL=0TOM:FI$(K+1,L)=FI$(K,L):NEXT
9050 K=K-1:IFK>0THEN9030
9060 FORL=0TOM:FI$(K+1,L)=B$(L):NEXT
9070 NEXTJ:RETURN
10000 GOSUB12000
10005 IFA$="E"THENPRINT"":PRINT"":RETURN
10010 PRINT"":PRINT"":GOSUB13000:PRINT@P+18,"** RECORDING DATA **":PRINT""
10020 PRINT#-1,M,G,N
10030 IFM=5THENFORI=1TON:PRINT#-1,FI$(I,0),FI$(I,1),FI$(I,2),FI$(I,3),FI$(I,4),FI$(I,
    5):NEXT:GOTO10050
10040 FORI=1TON:PRINT#-1,FI$(I,0),FI$(I,1),FI$(I,2),FI$(I,3),FI$(I,4),FI$(I,5),FI$(I,
    6):NEXT
```

(Program Listing, cont.)

```
10050 PRINT#-1, "", "", "", "", "", "", "": RETURN
11000 PRINT#270, "": GOSUB7000: PRINT#590, "": GOSUB6000: P=614: GOSUB3000: RETURN
12000 PRINT "SETUP THE RECORDER AND PRESS ENTER WHEN READY. ": PRINT "TO EXIT, PRESS E":
    GOSUB8000: RETURN
13000 P=(256*PEEK(16417)+PEEK(16416))-15359: RETURN
14000 PRINT "": GOSUB12000: IFA$="E" THEN RETURN
14010 PRINT "": PRINT "": GOSUB13000: PRINT#P+18, "** READING DATA **": PRINT "
14015 I=M: J=G: K=N
14020 INPUT#-1, M, G, N
14030 IF N>H THEN 14090: ELSE IF G<0 AND G<>1 THEN 14090: ELSE IF M<0 AND M<>6 THEN 14090
14040 N=1: IF M=6 THEN 14070
14050 INPUT#-1, FI$(N, 0), FI$(N, 1), FI$(N, 2), FI$(N, 3), FI$(N, 4), FI$(N, 5)
14060 IF FI$(N, 0)<>" " THEN N=N+1: GOTO14050: ELSE N=N-1: POKE16553, 255: RETURN
14070 INPUT#-1, FI$(N, 0), FI$(N, 1), FI$(N, 2), FI$(N, 3), FI$(N, 4), FI$(N, 5), FI$(N, 6)
14080 IF FI$(N, 0)<>" " THEN N=N+1: GOTO14070: ELSE N=N-1: POKE16553, 255: RETURN
14090 PRINT "": GOSUB13000: PRINT#P+18, "** TAPE READ ERROR **": M=I: G=J: N=K: GOTO14000
15000 FOR J=0 TO M: FI$(I, J)=B$(J): NEXT: RETURN
16000 CLS: PRINT#340, "** RECOVERING SPACE **": I=1
16010 IF FI$(I, 0)=" " THEN IF I=N THEN N=N-1: RETURN:
    ELSE FOR J=I TO N-1: FOR K=0 TO M: FI$(J
        , K)=FI$(J+1, K): NEXT: NEXT:
        N=N-1: I=I-1
16020 I=I+1: IF I>N THEN RETURN: ELSE 16010
17000 PRINT "  NAME--": IF G=1 THEN PRINT "  TEL---" ELSE PRINT "  COMP--"
17010 PRINT "  ADDR--": PRINT "  CITY--": PRINT "  ST----": PRINT "  ZIP---": IF M=6 THEN PR
    INT "  SEL---"
17020 RETURN
18000 IF B$(0)=" " THEN 19000
18010 IF B$(0)=LEFT$(FI$(I, 5), LEN(B$(0))) THEN 19000 ELSE RETURN
19000 K=1
19001 IF MID$(FI$(I, 0), K, 1)=" " THEN IF K=1 THEN RETURN: ELSE LPRINT MID$(FI$(I, 0), K+1), " "; LE
    FT$(FI$(I, 0), K-1): GOTO19005
19002 K=K+1: IF K<=LEN(FI$(I, 0)) THEN 19001 ELSE LPRINT FI$(I, 0)
19005 IF E=1 THEN LPRINT FI$(I, 1)
19010 LPRINT FI$(I, 2): LPRINT FI$(I, 3), "    "; FI$(I, 4), "-"; FI$(I, 5): LPRINT CHR$(138): LPR
    INT CHR$(138)
19015 IF E=0 THEN LPRINT CHR$(138): ELSE IF FI$(I, 1)=" " THEN LPRINT CHR$(138)
19020 RETURN
20000 DATA A, R, S, L, P, W
```
